

Przegląd funkcji Excela

– funkcje logiczne

Mariusz Jankowski

autor strony internetowej
poświęconej Excelowi
i programowaniu w VBA;

Bogdan Gilarski

właściciel firmy szkoleniowej
Perfect And Practical;

Pytania: czytelnicy.controlling@infor.pl

Funkcje logiczne są najmniej liczną kategorią funkcji Excela – w ich skład wchodzi tylko sześć funkcji (Excel 2003), począwszy od wersji 2007 zestaw ten powiększył się o JEŻELI.BŁĄD(). Są to funkcje niezwykle użyteczne, a często nawet niezastąpione w codziennej pracy. Trudno wyobrazić sobie poważniejszy arkusz kalkulacyjny np. bez wbudowanej funkcji JEŻELI().

Funkcje logiczne służą przede wszystkim do testowania warunków, które powinny spełniać wartości zapisane w komórkach arkusza. Rzadko kiedy stosuje się je samodzielnie, najczęściej są one zagnieżdżane w formułach, pełniąc rolę argumentów dla innych funkcji (nierzadko zwracają w wyniku tablicę wartości liczbowych jako parametr dla funkcji matematycznych lub statystycznych).

PRAWDA()

Jest to funkcja bezargumentowa, zwracająca w wyniku zawsze wartość PRAWDA. Wartość logiczną PRAWDA można wpisać także bezpośrednio do komórki arkusza lub formuły. Jak podaje pomoc, funkcja ta została wprowadzona do Excela, aby zachować zgodność z pozostałymi arkuszami kalkulacyjnymi.

FAŁSZ()

Podobnie jak w przypadku funkcji PRAWDA() jest to funkcja bezargumentowa. W wyniku zwraca zawsze wartość FAŁSZ. Biorąc pod uwagę fakt, że wartość logiczną FAŁSZ można wpisać bezpośrednio do komórki arkusza lub formuły, funkcja ta podobnie jak poprzednia wydaje się absolutnie zbędna w Excelu.

JEŻELI()

Funkcja sprawdza warunek logiczny podany w pierwszym argumentie funkcji. Jeżeli jest on spełniony (PRAWDA), w wyniku zwracana jest

wartość drugiego argumentu, a jeżeli nie (FAŁSZ) – wartość trzeciego argumentu.

Składnia funkcji JEŻELI () wygląda następująco:

- *test_logiczny* – sprawdzany warunek logiczny. Jest to wartość lub wyrażenie, którego wynik można oszacować jako wartość logiczną (PRAWDA lub FAŁSZ). W warunku logicznym bardzo często wykorzystywane są operatory porównania, takie jak: =, >, <, >=, <=, <>, za pomocą których pokazywane są wartości komórek. Na przykład, A1 = 5 jest wyrażeniem logicznym; jeżeli wartością umieszczoną w komórce A1 jest faktycznie cyfra 5, to wyrażenie to zostanie oszacowane jako PRAWDA, w każdym innym przypadku wyrażenie zostanie oszacowane jako FAŁSZ;

- *wartość_jeżeli_prawda* – drugi argument funkcji jest tak naprawdę wartością zwracaną w sytuacji, gdy pierwszy argument *test_logiczny* jest spełniony. Jeżeli naszym testem logicznym jest wyrażenie A1 = 5 i w komórce A1 faktycznie znajduje się cyfra 5, to wynikiem funkcji (wartością drugiego argumentu) może być komunikat tekstowy „w tej komórce znajduje się cyfra 5”. Jest to argument opcjonalny, który może być umieszczony wewnątrz funkcji lub pominięty poprzez wstawienie dwóch średników po argumentie pierwszym. Pominięcie wskazuje, że jest to argument domyślny, o wartości wynikowej PRAWDA;

● *wartość_jeżeli_fałsz* – trzeci argument funkcji jest wartością zwracaną w sytuacji, gdy pierwszy argument *test_logiczny* zwraca w wyniku wartość FAŁSZ. Jeżeli naszym testem logicznym w dalszym ciągu pozostaje wyrażenie $A1 = 5$, natomiast w komórce A1 znajduje się inna wartość (liczba, tekst, błąd, wartość logiczna) lub komórka jest pusta, to wynikiem funkcji (wartością trzeciego argumentu) może być komunikat tekstowy „wartość tej komórki jest różna od cyfry 5”. Pozostawienie tego argumentu pustym spowoduje natomiast zwrócenie w wyniku wartości FAŁSZ. Należy wyraźnie podkreślić, że nie można pominąć dwóch argumentów opcjonalnych, przynajmniej jeden z nich powinien być określony;

● widok 1 pokazuje wynik działania funkcji JEŻELI() w zależności od wartości komórki po lewej stronie od formuły. Jak widać, tylko w drugim przypadku wyrażenie logiczne jest prawdziwe. Warto zaznaczyć, że Excel nie stosuje tutaj automatycznej konwersji i prawidłowo

wo rozróżnia wartości liczbowe od tekstu, dlatego cyfra 5 poprzedzona apostrofem jest traktowana jak tekst (warunek logiczny zwraca wartość FAŁSZ, formuła natomiast – tekst zawarty w trzecim argumente funkcji *wartość_jeżeli_fałsz*).

Widok 1

Przykład zastosowania funkcji JEŻELI()

	A	B	C
1	Przykład zastosowania funkcji JEŻELI()		
2			
3	5,1	wartość tej komórki jest różna od cyfry 5	
4	5	w tej komórce znajduje się cyfra 5	
5		wartość tej komórki jest różna od cyfry 5	
6	5	wartość tej komórki jest różna od cyfry 5	
7	tekst	wartość tej komórki jest różna od cyfry 5	
8			

Formuła znajdująca się w komórce B3 wygląda następująco:
 =JEŻELI (\$A3=5;"w tej komórce znajduje się cyfra 5"; "wartość tej komórki jest różna od cyfry 5")

Przykład 1

Warunek logiczny z funkcją statystyczną

Jak napisaliśmy wcześniej, w przypadku testów logicznych funkcji JEŻELI() bardzo często korzysta się z operatorów porównania. Tym razem oprócz operatora >= wykorzystana jest funkcja statystyczna ŚREDNIA(). Zadanie polega na tym, aby wyróżnić uczniów, którzy uzyskali średnią co najmniej 5,00 z przedmiotów matematyczno-przyrodniczych – osoby takie kwalifikują się do przyznania szkolnego stypendium za wyniki w nauce.

Widok 2. Warunek logiczny z funkcją statystyczną

	A	B	C	D	E	F	G
1	Warunek logiczny z funkcją statystyczną()						
2							
3	Uczeń	Matematyka	Fizyka	Chemia	Biologia	STYPENDIUM	
4	A. Marecki	5	4	6	5	tak - 5,00	
5	B. Grabarczyk	6	4	5	4	nie - 4,75	
6	K. Wójcik	5	4	6	4	nie - 4,75	
7	L. Nowotny	4	6	5	6	tak - 5,25	
8	M. Darecki	6	5	5	5	tak - 5,25	
9	A. Kowalska	6	4	4	5	nie - 4,75	
10	S. Bednarska	6	5	5	4	tak - 5,00	
11							

Formuła w komórce F4 sprawdza średnią z czterech przedmiotów i wygląda tak:

```
=JEŻELI(ŚREDNIA($B4:$E4)>=5;"tak - "&TEKST(ŚREDNIA($B4:$E4);"0,00");
"nie - "&TEKST(ŚREDNIA($B4:$E4);"0,00"))
```

Kluczowy w tym przypadku jest warunek logiczny, czyli sprawdzenie, czy średnia z czterech przedmiotów jest wyższa lub równa 5,00. Jeżeli tak się dzieje, formuła zwraca w wyniku tekst „tak -” i pokazuje średnią dla ucznia, w przeciwnym wypadku zwraca tekst „nie -” wraz ze średnią. Formuła została nieco skomplikowana w celach wizualnych, zdecydowanie lepszą praktyką jest utworzenie dwóch osobnych kolumn: na średnią i informację o stypendium (w takim wypadku możliwe jest np. sortowanie według średniej).

Przykład 2

Zagnieżdżenie funkcji JEŻELI()

Niektóre problemy logiczne mogą być rozwiązane dopiero za pomocą kilkakrotnego zagnieżdżenia funkcji JEŻELI() np. w sytuacji, gdy trzeba przydzielić ocenę zależnie od wartości przedziału, w jakim znajduje się wynik. **Widok 3** pokazuje tabelę, w której przypisano uczniom ocenę na podstawie ilości punktów uzyskanych z egzaminu. Formuła dla komórki C4 to:

```
=JEŻELI($B4>18;"celujący";JEŻELI($B4>16;"bardzo dobry";
JEŻELI($B4>13;"dobry";JEŻELI($B4>10;"dostateczny";
JEŻELI($B4>8;"mierny";"niedostateczny"))))
```

Widok 3. Zagnieżdżenie funkcji JEŻELI()

	A	B	C	D	E	F	G	H
1	Zagnieżdżenie funkcji JEŻELI()							
2								
3	Uczeń	Punkty	Ocena		Przedział	Ocena		
4	W. Kowalska	16	dobry		0	8	niedostateczny	
5	M. Panek	13	dostateczny		9	10	mierny	
6	A. Błozowska	17	bardzo dobry		11	13	dostateczny	
7	L. Iwan	9	mierny		14	16	dobry	
8	M. Słaboń	19	celujący		17	18	bardzo dobry	
9	D. Wojtczak	5	niedostateczny		19	20	celujący	
10								

Warto zwrócić uwagę na kilka istotnych rzeczy:

- Funkcja JEŻELI() może być zagnieżdżana maksymalnie do 7 razy w formule (Excel 2003). W Excelu 2007 można zagnieżdżyć ją aż 64 razy, jednak formuła taka nie będzie działać prawidłowo we wcześniejszych wersjach MS Excel, dlatego sugeruje się wykorzystanie maksymalnie 7 zagnieżdżeń w pojedynczej formule. W przypadku naszej formuły mamy do czynienia z pięciokrotnym użyciem zagnieżdżanej sukcesywnie funkcji JEŻELI().
- Druga instrukcja funkcji JEŻELI() jest jednocześnie argumentem *wartość_jeżeli_fałsz* dla pierwszej instrukcji funkcji (**widok 4**). Analogicznie trzecia instrukcja funkcji JEŻELI() jest argumentem *wartość_jeżeli_fałsz* dla drugiej instrukcji itd.

Widok 4. Zagnieżdżenie warunków

Argumenty funkcji

JEŻELI

Test_logiczny: \$B4>18 = FAŁSZ

Wartość_jeżeli_prawda: "celujący" = "celujący"

Wartość_jeżeli_fałsz: JEŻELI(\$B4>16;"bardzo dobry";JEŻELI(\$B4>13;"dobry";JEŻELI(\$B4>10;"dostateczny";JEŻELI(\$B4>8;"mierny";"niedostateczny"))) = "dobry"

Sprawdza, czy warunek jest spełniony, i zwraca jedną wartość, jeśli PRAWDA, a drugą wartość, jeśli FAŁSZ.

Test_logiczny - dowolna wartość lub wyrażenie, które można oszacować jako wartości PRAWDA albo FAŁSZ.

Wynik formuły = dobry

Pomoc dotycząca tej funkcji

OK Anuluj

- Formuła sprawdza pierwszy warunek logiczny i przerywa działanie w momencie, gdy zwraca on wartość PRAWDA (w wyniku zwraca wartość drugiego argumentu). Jeżeli zwracaną wartością jest FAŁSZ – formuła przechodzi do drugiego warunku itd. W naszym przykładzie (liczba punktów = 16) pierwszy warunek \$B4>18 jest

nieprawdziwy, drugi \$B4>16 także zwraca FAŁSZ. Dopiero trzeci \$B4>13 jest spełniony – wynikiem formuły jest więc słowo „dobry”.

- Dla liczby punktów mniejszej niż 9 każdy z pięciu warunków zwraca FAŁSZ. W takiej sytuacji (w innym przypadku) formuła zwraca wyraz „niedostateczny”.

Przykład 3

Funkcja JEŻELI() operująca na tablicy wartości

Tworząc w pierwszym parametrze funkcji określony test logiczny, nie musimy ograniczać się do podawania pojedynczej komórki arkusza. Funkcja JEŻELI() zezwala na podanie zakresu komórek (który jest przetwarzany w pamięci na jedno- lub dwuwymiarową tablicę) lub ręczne wpisanie tablicy. Stosując operator porównania lub konkretną funkcję Excela, możemy sprawdzić warunek dla każdej komórki (dla każdego elementu tablicy) oddzielnie. **Widok 5** przedstawia zakres A3:E10, który zawiera dane różnego typu: liczby, tekst, błędy, wartości logiczne. Zadanie polega na utworzeniu jednej formuły, która zliczy wszystkie wpisy, które nie są liczbami.

Widok 5. Funkcja JEŻELI() operująca na tablicy wartości

Rozwiązaniem zadania jest formuła tablicowa (Ctrl+Shift+Enter):

=SUMA(JEŻELI(NIE(CZY.LICZBA(dane));1;0))

- Formuła zwraca prawidłowy wynik 18 (komórki puste i wartości logiczne nie są liczbami).

- Najpierw formuła sprawdza, czy wszystkie komórki z zakresu Dane są liczbami. Tworzy tablicę złożoną z 40 elementów typu PRAWDA/FALSZ.

- Ponieważ interesują nas wpisy, które nie są liczbami, stosujemy przeczenie NIE(), które zamienia wartość PRAWDA na FALSZ i odwrotnie.

- Następnie każdej wartości typu PRAWDA funkcja JEŻELI() przypisuje cyfrę 1, każdej wartości typu FALSZ wartość 0. Efektem takiego działania jest tablica złożona z jedynek i zer.

- Mając taką tablicę, wystarczy dodać do siebie wszystkie jedynek, co czyni funkcja SUMA().

- Jak widać, Excel traktuje daty jako liczby (zarówno te posiadające standardowy format, jak i te wyświetlające tekst), dlatego nie zostały one wyróżnione.

- Taka konstrukcja formuły świetnie sprawdza się do przeprowadzania różnych operacji matematycznych lub statystycznych na wybranej zmiennej. Należy jednak pamiętać, aby w przypadku obliczania średniej zamienić wartości typu FALSZ na puste ciągi, a nie na zera, które także są wliczane do średniej i przez to generują nieprawidłowe wyniki.

- Zadanie to może być także zrealizowane za pomocą krótszej formuły tablicowej:

=SUMA(JEŻELI(CZY.LICZBA(dane);0;1)), a także bez użycia funkcji JEŻELI():
=SUMA(NIE(CZY.LICZBA(dane))*1)

	A	B	C	D	E	F
1	Funkcja JEŻELI() operująca na tablicy wartości					
2						
3	132	2343	5435	tekst	FALSZ	
4	4234	4234		wrzesień		
5	4234	34	3	#DZIEL/0!	PRAWDA	
6	4324		3	2010-09-16	3	3
7	PRAWDA	#N/D!	3		324	424
8	3		tekst		3	3
9	10:51	FALSZ	3	PRAWDA	555	
10	3432	433	234			12
11						
12		dane	=Arkusz4!\$A\$3:\$E\$10			
13						

ORAZ()

Funkcja ORAZ() współdziała z funkcją JEŻELI(), rozszerzając jej działanie. JEŻELI() w czystej postaci operuje tylko jednym warunkiem logicznym, natomiast ORAZ() pozwala na wprowadzenie maksymalnie 30 warunków logicznych. Aby funkcja zwracała w wyniku wartość PRAWDA, konieczne jest, aby każdy z warunków, argumentów funkcji, był spełniony. Jeżeli chociaż jeden z przyjętych warunków ma wartość FALSZ, funkcja zwraca w wyniku FALSZ.

ORAZ() jest więc, inaczej mówiąc, iloczynem (koniunkcją) warunków. Należy pilnować, aby wszystkie argumenty funkcji były zawsze wartościami logicznymi, ponieważ używając innych typów wartości, bardzo łatwo można wygenerować błąd. Używanie argumentów pustych i tekstowych jest dopuszczalne tylko w szczególnie uzasadnionych przypadkach (jeżeli drugi z argumentów jest odwołaniem do pustej komórki lub takiej, która zawiera tekst, formuła będzie działać, dane wpisane do parametrów ręcznie generują natomiast błąd #ARG!).

Przykład 4

Koniunkcja warunków

Widok 6 przedstawia prostą tabelkę, na bazie której w sposób bardzo użyteczny można wykorzystać funkcję ORAZ(). Problem dotyczy selekcji kandydatów na stanowisko *Analityka* w naszej firmie. W sumie zgłosiło się siedem osób, które poddano testom z różnych dziedzin. Założenie jest takie, że interesują nas osoby, które bardzo dobrze znają język angielski, Excela i uzyskali co najmniej 15 punktów z naszego testu sprawdzającego wiedzę z zakresu controllingu. Jeżeli kandydat spełnia wszystkie te warunki, wówczas zostaje dopuszczony do drugiego etapu rekrutacji; w przeciwnym razie odpada.

Widok 6. Koniunkcja warunków

	A	B	C	D	E	F	G	H
1	Selekcja kandydatów na stanowisko Analityka - wyniki							
2								
3	Kandydat	Test (pkt.)	J. angielski	J. niemiecki	Excel	Access	Wynik	
4	W. Mirecki	15	nie	nie	tak	nie	Odpada	
5	A. Bandera	16	tak	nie	tak	tak	II etap	
6	K. Morawiec	11	nie	nie	nie	nie	Odpada	
7	M. Iwańczyk	16	tak	tak	tak	tak	II etap	
8	D. Maciąg	12	tak	nie	nie	nie	Odpada	
9	W. Lubomirski	16	nie	tak	tak	tak	Odpada	
10	C. Darecki	17	tak	nie	tak	nie	II etap	
11								

Formuła zawarta w komórce \$G4 wygląda następująco:

=JEŻELI(ORAZ(\$B4>14;\$C4="tak";\$E4="tak");"II etap";"Odpada")

- Jak widać, dla pierwszej osoby formuła zwraca tekst "Odpada" ponieważ nie został spełniony drugi z warunków – bardzo dobra znajomość języka angielskiego. Pomimo że dwa pozostałe warunki zostały spełnione – funkcja ORAZ() zwraca w wyniku FALSZ, co funkcja JEŻELI() zamienia na konkretny komunikat tekstowy.
- Do drugiego etapu zakwalifikowano trzy osoby, które spełniły wszystkie warunki.

Przykład 5

Alternatywa warunków

Widok 7 przedstawia tę samą tabelkę, ale z innymi wynikami. Tym razem rekrutujemy na stanowisko *Specjalisty ds. Bazy Danych*, dlatego nieco inne są kryteria, które decydują o kwalifikacji do następnego etapu. Kandydat tym razem musi spełniać dwa warunki: znać biegle Accessa i przynajmniej jeden z dwóch języków obcych.

Widok 7. Alternatywa warunków

	A	B	C	D	E	F	G	H
1	Selekcja kandydatów na stanowisko Specjalisty ds.. Bazy Danych - wyniki							
2								
3	Kandydat	Test (pkt.)	J. angielski	J. niemiecki	Excel	Access	Wynik	
4	W. Mirecki	15	nie	nie	tak	nie	Odpada	
5	A. Bandera	16	tak	nie	tak	tak	II etap	
6	K. Morawiec	11	nie	nie	nie	nie	Odpada	
7	M. Iwańczyk	16	tak	tak	tak	tak	II etap	
8	D. Maciąg	12	tak	nie	nie	nie	Odpada	
9	W. Lubomirski	16	nie	tak	tak	tak	II etap	
10	C. Darecki	17	tak	nie	tak	nie	Odpada	
11								

Formuła zawarta w komórce \$G4 wygląda następująco:

JEŻELI(ORAZ(\$F4="tak";LUB(\$C4="tak";\$D4="tak"));"II etap";"Odpada")

- W zasadzie mamy tutaj połączenie koniunkcji i alternatywy warunków. Koniunkcja polega na tym, że kandydat musi spełnić dwa warunki, alternatywa na tym, że wymagana jest dobra znajomość dowolnego języka obcego.
- Podobnie jak w przykładzie dotyczącym rekrutacji analityka, i tym razem do następnego etapu zakwalifikowały się trzy osoby. Każda osoba znająca biegle Accessa zna także przynajmniej jeden język obcy.

LUB()

Funkcja LUB() – podobnie jak funkcja ORAZ() – współdziała z funkcją JEŻELI(), pozwalając na tworzenie bardziej rozbudowanych testów. Aby funkcja zwracała w wyniku wartość PRAWDA, konieczne jest, aby przynajmniej jeden z warunków, argumentów funkcji, był spełniony. Jeżeli wszystkie z przyjętych warunków mają wartość FAŁSZ, funkcja zwraca w wyniku FAŁSZ. LUB() jest więc, inaczej mówiąc, sumą (alternatywą) warunków. Użycie argumentów pustych i tekstowych jest traktowane przez funkcję analogicznie jak w przypadku funkcji ORAZ().


Wnioski

Oprócz funkcji JEŻELI() wszystkie funkcje logiczne zwracają w wyniku wartości logiczne

typu PRAWDA/FAŁSZ. Są one niezastąpione przy testowaniu warunków, zarówno dla pojedynczej wartości (komórki) jak i tablicy wartości (zakresu komórek). Funkcja JEŻELI() może być wykorzystywana na wiele różnych sposobów: za pomocą zagnieżdżeń można sprawdzić nawet siedem różnych warunków; traktując poszczególne komórki zakresu jako parametr dla funkcji matematycznych, statystycznych lub informacyjnych, można dokonać obliczeń (np. zliczyć lub zsumować) tylko na tych komórkach, które spełniają określone kryteria. Funkcje ORAZ() i LUB() działają na pojedynczych komórkach i pozwalają rozbudowywać testy logiczne, przez co są one bardzo często wykorzystywane do tworzenia tzw. kryteriów formułowych (wyliczeniowych) w przypadku filtra zaawansowanego i funkcji baz danych. ■

ISC INTERNETOWY SERWIS CONTROLLINGU

www.isc.infor.pl



Internetowy Serwis Controllingu to pierwsza w Polsce w pełni profesjonalna platforma pozyskiwania wiedzy, niezbędna w pracy controllerów, analityków finansowych oraz menedżerów odpowiedzialnych za realizację funkcji controllingowych w przedsiębiorstwach. Zawiera kilka tysięcy praktycznych materiałów edukacyjnych (porad, artykułów, wyjaśnień).



Informacje i zamówienia: Biuro Obsługi Klienta, tel.: 22 761 30 30, 801 626 666, e-mail: bok@infor.pl